Pensieve header: The double and meta-double of the 2D pencil; continues pensieve://2017-04/.

```
SetDirectory["C:\\drorbn\\AcademicPensieve\\2017-05"];
```

## The 2D Lie BiAlgebra Pencil

I hope to stick to $G = e^{\eta g}$ and to $H = e^{\gamma h}$, where $[g, e] = \gamma e$ and $[h, f] = -\eta f$.
Also, $q\Delta_{12}(g, G, e, h, H, f) = (g_1 + g_2,\ G_1 G_2,\ e_1 + G_1 e_2,\ h_1 + h_2,\ H_1 H_2,\ f_1 H_2 + f_2)$.
Also, $(g, e)$ and $(h, f)$ are dual bases.

```
AlgebraAtom = g | G[_] | e | h | H[_] | f;
$PBWRule = {G[_] → 1, g → 2, e → 3, H[_] → 4, h → 5, f → 6};
```

```
B[g, e] = γ e; B[e, G[n_]] = (e^(-n γ η) - 1) U[G[n], e]; B[g, G[_]] = 0;
B[h, f] = -η f; B[f, H[n_]] = (e^(n γ η) - 1) U[H[n], f]; B[h, H[_]] = 0;
```

## UEA with provisional modification

This section is based on penseive://Projects/UEA/.

```
B[0, _] = 0; B[_, 0] = 0;
B[c_ * x : AlgebraAtom, y_] := Expand[c B[x, y]];
B[y_, c_ * x : AlgebraAtom] := Expand[c B[y, x]];
B[x_Plus, y_] := B[#, y] & /@ x;
B[x_, y_Plus] := B[x, #] & /@ y;
B[x_, x_] = 0;
B[y_, x_] := Expand[-B[x, y]];
```

```
x_ ≤ y_ := OrderedQ[{x, y} /. $PBWRule]; x_ < y_ := ! OrderedQ[{y, x} /. $PBWRule];
UU_i_[1] := U_i[];
UU_i_[x_[n_]^p_] := U_i[x[n p]];
UU_i_[x_^p_] := UU_i @@ Table[x, {p}];
UU_i_[ℰ_] := ℰ /. {
    U[xs__] :> U_i[xs],
    x : AlgebraAtom :> U_i[x]
   };
UU_i_[x_, xs__] := UU_t1[x] UU_t2[xs] // Expand // m_{t1,t2→i};
USimp[ℰ_] := Collect[ℰ, Times[U_[__] ..], Expand];
USimp[ℰ_] := Expand[ℰ];
```

```
m_s__[0] = 0;
m_s__[x_Plus] := m_s /@ x;
m_s__[sd_SeriesData] := MapAt[m_s, sd, {3, All}];
m_i_→j_[ℰ_] := ℰ /. U_i → U_j;
```

```
m_{i_,j_→k_}[c_. U_{i_}[x___] U_{j_}[]] := c U_k[x];
m_{i_,j_→k_}[c_. U_{i_}[] U_{j_}[y___]] := c U_k[y];
m_{i_,j_→k_}[c_. U_{i_}[xx___, x_[n1_]] U_{j_}[x_[n2_], yy___]] :=
  USimp[c If[n1 + n2 == 0, U_i[xx] U_j[yy], U_i[xx, x[n1 + n2]] U_j[yy]] // m_{i,j→k}];
m_{i_,j_→k_}[c_. U_{i_}[xx___, x_] U_{j_}[y_, yy___]] := If[x ≤ y,
   c U_k[xx, x, y, yy],
   ((U_i[xx] (U_j[y, x] + UU_j[B[x, y]]) // Expand // m_{i,j→i}) U_j[yy] // Expand // m_{i,j→k})
     c // USimp
  ];
```

```
Supp[ℰ_] := Union@Cases[{ℰ}, U_{i_}[___] ⧴ i, ∞];
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
x_ ** y_ := Module[{is = Supp[x] ⋂ Supp[y], σ, z},
   z = x; Do[z = m_{i→σ@i}[z], {i, is}];
   z = Expand[y z]; Do[z = m_{σ@i,i→i}[z], {i, is}]; z];
UB[x_, y_] := USimp[x ** y - y ** x];
```

```
𝕆[specs_, sd_SeriesData] := MapAt[𝕆[specs, #] &, sd, {3, All}];
𝕆[specs_, poly_] := Module[{rules, vars, elems},
  rules = Union @@ Cases[{specs}, U_[u___] ⧴ Cases[{u}, r_Rule], ∞];
  vars = First /@ rules; elems = Last /@ rules;
  USimp@Total[CoefficientRules[poly, vars] /. (ps_ → c_) ⧴ c (
       specs /. MapThread[((#1 → _) ⧴ #3^#2) &, {vars, ps, elems}] /. U_{i_} ⧴ UU_i
      )]
 ]
```

## The 2D Lie BiAlgebra Pencil, Testing

```
𝕆[U_1[x → g], Normal@Series[ⅇ^{η x}, {η, 0, 5}]]
```

$$U_1[\,] + \eta\, U_1[g] + \frac{1}{2}\eta^2 U_1[g, g] + \frac{1}{6}\eta^3 U_1[g, g, g] + \frac{1}{24}\eta^4 U_1[g, g, g, g] + \frac{1}{120}\eta^5 U_1[g, g, g, g, g]$$

```
With[{G = 𝕆[U_1[x → g], Series[ⅇ^{η x}, {η, 0, 5}]]}, UB[U_1[e], G] - (ⅇ^{-γ η} - 1) G ** U_1[e]]
```

$O[\eta]^6$

```
B[e, G[3]]
```

$\left(-1 + e^{-3\gamma\eta}\right) U[G[3], e]$

```
With[{H = 𝕆[U_1[x → h], Series[ⅇ^{γ x}, {γ, 0, 5}]]}, UB[U_1[f], H] - (ⅇ^{γ η} - 1) H ** U_1[f]]
```

$O[\gamma]^6$

## The Co-Product and Co-Associativity

```
qΔ_{i_→j_,k_}[ℰ_] := USimp@Module[{tj, tk}, ℰ /. {
    U_i[] → U_j[] U_k[],
    U_i[g, xs___] :→
     (USimp[(U_j[g] U_k[] + U_j[] U_k[g]) qΔ_{i→tj,tk}[U_i[xs]]] // m_{j,tj→j} // m_{k,tk→k}),
    U_i[G[n_], xs___] :→ (USimp[U_j[G[n]] U_k[G[n]] qΔ_{i→tj,tk}[U_i[xs]]] //
       m_{j,tj→j} // m_{k,tk→k}),
    U_i[e, xs___] :→ (USimp[(U_j[e] U_k[G[1]] + U_j[] U_k[e]) qΔ_{i→tj,tk}[U_i[xs]]] //
       m_{j,tj→j} // m_{k,tk→k}),
    U_i[h, xs___] :→ (USimp[(U_j[h] U_k[] + U_j[] U_k[h]) qΔ_{i→tj,tk}[U_i[xs]]] //
       m_{j,tj→j} // m_{k,tk→k}),
    U_i[H[n_], xs___] :→ (USimp[U_j[H[n]] U_k[H[n]] qΔ_{i→tj,tk}[U_i[xs]]] //
       m_{j,tj→j} // m_{k,tk→k}),
    U_i[f, xs___] :→ (USimp[(U_j[f] U_k[] + U_j[H[1]] U_k[f]) qΔ_{i→tj,tk}[U_i[xs]]] //
       m_{j,tj→j} // m_{k,tk→k})
   }];
```

```
qΔ_{i_→j_,k_,l__}[ℰ_] := ℰ // qΔ_{i→j,k} // qΔ_{k→k,l}
```

```
U_1[e] // qΔ_{1→1,2}
```

```
U_1[] U_2[e] + U_1[e] U_2[G[1]]
```

```
{lhs = U_1[e] // qΔ_{1→1,2} // qΔ_{2→2,3}, rhs = U_1[e] // qΔ_{1→1,3} // qΔ_{1→1,2}, lhs == rhs}
```

```
{U_1[] U_2[] U_3[e] + U_1[] U_2[e] U_3[G[1]] + U_1[e] U_2[G[1]] U_3[G[1]],
 U_1[] U_2[] U_3[e] + U_1[] U_2[e] U_3[G[1]] + U_1[e] U_2[G[1]] U_3[G[1]], True}
```

```
U_1[f] // qΔ_{1→1,2}
```

```
U_1[f] U_2[] + U_1[H[1]] U_2[f]
```

```
{lhs = U_1[f] // qΔ_{1→1,2} // qΔ_{2→2,3}, rhs = U_1[f] // qΔ_{1→1,3} // qΔ_{1→1,2}, lhs == rhs}
```

```
{U_1[f] U_2[] U_3[] + U_1[H[1]] U_2[f] U_3[] + U_1[H[1]] U_2[H[1]] U_3[f],
 U_1[f] U_2[] U_3[] + U_1[H[1]] U_2[f] U_3[] + U_1[H[1]] U_2[H[1]] U_3[f], True}
```

## The Antipode

Why o why this annoyance of left-vs-right?

```
S[g] = -g; S[G[n_]] := G[-n]; S[e]  = -e^{γη} U[G[-1], e];
S[h] = -h; S[H[n_]] := H[-n]; S[f]  = -U[H[-1], f];
S_i_[ℰ_] := Module[{ti}, USimp[
   ℰ /. U_i[x_, xs___] :→ m_{ti,i→i}[Expand[UU_i[S[x]] S_{ti}[U_{ti}[xs]]]]
  ]];
```

```
{lhs = S1[U1[e]], rhs = -U1[e] ** U1[G[-1]], lhs == rhs}
```

$\{-e^{\gamma\eta}U_1[G[-1], e], -e^{\gamma\eta}U_1[G[-1], e], \text{True}\}$

```
U1[e] // S1 // S1
```

$e^{\gamma\eta}U_1[e]$

```
U1[f] // S1 // S1
```

$e^{\gamma\eta}U_1[f]$

```
S1[U1[g, G[3], e, e]]
```

$2e^{9\gamma\eta}\gamma U_1[G[-5], e, e] - e^{9\gamma\eta}U_1[G[-5], g, e, e]$

```
U1[g, G[3], e, e] // qΔ1→1,2
```

$U_1[G[3], g, e, e]U_2[G[5]] + U_1[G[3], g, e]U_2[G[4], e] + e^{-\gamma\eta}U_1[G[3], g, e]U_2[G[4], e] + U_1[G[3], e, e]U_2[G[5], g] + U_1[G[3], g]U_2[G[3], e, e] + U_1[G[3], e]U_2[G[4], g, e] + e^{-\gamma\eta}U_1[G[3], e]U_2[G[4], g, e] + U_1[G[3]]U_2[G[3], g, e, e]$

```
U1[g, G[3], e, e] // qΔ1→1,2 // S2
```

$U_1[G[3], g, e, e]U_2[G[-5]] - e^{4\gamma\eta}\gamma U_1[G[3], e]U_2[G[-5], e] - e^{5\gamma\eta}\gamma U_1[G[3], e]U_2[G[-5], e] - e^{4\gamma\eta}U_1[G[3], g, e]U_2[G[-5], e] - e^{5\gamma\eta}U_1[G[3], g, e]U_2[G[-5], e] - U_1[G[3], e, e]U_2[G[-5], g] + 2e^{9\gamma\eta}\gamma U_1[G[3]]U_2[G[-5], e, e] + e^{9\gamma\eta}U_1[G[3], g]U_2[G[-5], e, e] + e^{4\gamma\eta}U_1[G[3], e]U_2[G[-5], g, e] + e^{5\gamma\eta}U_1[G[3], e]U_2[G[-5], g, e] - e^{9\gamma\eta}U_1[G[3]]U_2[G[-5], g, e, e]$

```
test = U1[g, G[3], e, e];
{test // qΔ1→1,2 // S2 // m1,2→1, test // qΔ1→1,2 // S2 // m2,1→1,
 test // qΔ1→1,2 // S1 // m1,2→1, test // qΔ1→1,2 // S1 // m2,1→1}
```

$\{0, 0, 0, 2e^{6\gamma\eta}\gamma U_1[e, e] - 2e^{7\gamma\eta}\gamma U_1[e, e] - 2e^{8\gamma\eta}\gamma U_1[e, e] + 2e^{9\gamma\eta}\gamma U_1[e, e]\}$

```
test = U1[h, H[3], f, f];
{test // qΔ1→1,2 // S2 // m1,2→1, test // qΔ1→1,2 // S2 // m2,1→1,
 test // qΔ1→1,2 // S1 // m1,2→1, test // qΔ1→1,2 // S1 // m2,1→1}
```

$\{0, 0, 0, -2e^{-10\gamma\eta}\eta U_1[H[-2], f, f] + 2e^{-9\gamma\eta}\eta U_1[H[-2], f, f] + 2e^{-8\gamma\eta}\eta U_1[H[-2], f, f] - 2e^{-7\gamma\eta}\eta U_1[H[-2], f, f]\}$

## The Pairing at Lie-Level and Compatibilities

```
P[U[], U[]] = 1;
P[U[], U[H[_]]] = P[U[G[_]], U[]] = 1;
P[U[], U[__]] = P[U[__], U[]] = 0;
( P[U[g], U[h]] = 1         P[U[g], U[H[n_]]] = n γ          P[U[g], U[f]] = 0
  P[U[G[n_]], U[h]] = n η   P[U[G[n_]], U[H[m_]]] = e^(n m η γ)   P[U[G[_]], U[f]] = 0
  P[U[e], U[h]] = 0         P[U[e], U[H[_]]] = 0             P[U[e], U[f]] = 1 );
```

```
P_{i_,j_}[ℰ_] := USimp[ℰ /. U_i[xs___] U_j[ys___] → P[U[xs], U[ys]]];
```

```
t = Ui[g] Uj[e] Uk[f];
{mi,j→i[t] - mj,i→i[t], qΔk→k,l[t] - qΔk→l,k[t]}
```

```
{γ Ui[e] Uk[f], Ui[g] Uj[e] Uk[f] Ul[] - Ui[g] Uj[e] Uk[] Ul[f] +
  Ui[g] Uj[e] Uk[H[1]] Ul[f] - Ui[g] Uj[e] Uk[f] Ul[H[1]]}
```

```
t = Ui[g] Uj[e] Uk[f];
{(mi,j→i[t] - mj,i→i[t]) // Pi,k, (qΔk→k,l[t] - qΔk→l,k[t]) // Pi,k // Pj,l}
```

```
{γ, γ}
```

```
Table[t = Ui[xi] Uj[xj] Uk[yk];
 {(mi,j→i[t] - mj,i→i[t]) // Pi,k, (qΔk→k,l[t] - qΔk→l,k[t]) // Pi,k // Pj,l},
 {xi, {g, e}}, {xj, {g, e}}, {yk, {h, f}}]
```

```
{{{{0, 0}, {0, 0}}, {{0, 0}, {γ, γ}}}, {{{0, 0}, {-γ, -γ}}, {{0, 0}, {0, 0}}}}
```

```
Table[t = Ui[xi] Uk[yk] Ul[yl];
 {(qΔi→i,j[t] - qΔi→j,i[t]) // Pi,k // Pj,l, (mk,l→k[t] - ml,k→k[t]) // Pi,k},
 {xi, {g, e}}, {yk, {h, f}}, {yl, {h, f}}]
```

```
{{{{0, 0}, {0, 0}}, {{0, 0}, {0, 0}}}, {{{0, 0}, {-η, -η}}, {{η, η}, {0, 0}}}}
```

## General Pairings

The pairing sequence: ⟨one,one⟩ (above), ⟨many,one⟩, ⟨many,many⟩.

```
P[U[x_, xs__], U[y_]] := P[U[x, xs], U[y]] =
   Module[{i, j, k, l}, USimp[Ui[x] Uj[xs] qΔk→k,l[Uk[y]]] // Pi,k // Pj,l];
P[U[xs__], U[y_, ys__]] := P[U[xs], U[y, ys]] =
   Module[{i, j, k, l}, USimp[qΔi→i,j[Ui[xs]] Uk[y] Ul[ys]] // Pi,k // Pj,l];
```

```
{P[U[g, e], U[h]], P[U[g, e], U[f]], P[U[e, e], U[f]]}
```

```
{0, γ, 0}
```

```
P[U[e], U[f, f]]
```

```
0
```

```
P[U[e, e], U[f, f]]
```

$1 + e^{\gamma\eta}$

```
lhs = Factor@Table[P[U@@Table[e, {n}], U@@Table[f, {n}]], {n, 7}]
```

$$\{1,\ 1+e^{\gamma\eta},\ (1+e^{\gamma\eta})(1+e^{\gamma\eta}+e^{2\gamma\eta}),\ (1+e^{\gamma\eta})^2(1+e^{2\gamma\eta})(1+e^{\gamma\eta}+e^{2\gamma\eta}),$$
$$(1+e^{\gamma\eta})^2(1+e^{2\gamma\eta})(1+e^{\gamma\eta}+e^{2\gamma\eta})(1+e^{\gamma\eta}+e^{2\gamma\eta}+e^{3\gamma\eta}+e^{4\gamma\eta}),$$
$$(1+e^{\gamma\eta})^3(1+e^{2\gamma\eta})(1-e^{\gamma\eta}+e^{2\gamma\eta})(1+e^{\gamma\eta}+e^{2\gamma\eta})^2(1+e^{\gamma\eta}+e^{2\gamma\eta}+e^{3\gamma\eta}+e^{4\gamma\eta}),$$
$$(1+e^{\gamma\eta})^3(1+e^{2\gamma\eta})(1-e^{\gamma\eta}+e^{2\gamma\eta})(1+e^{\gamma\eta}+e^{2\gamma\eta})^2$$
$$(1+e^{\gamma\eta}+e^{2\gamma\eta}+e^{3\gamma\eta}+e^{4\gamma\eta})(1+e^{\gamma\eta}+e^{2\gamma\eta}+e^{3\gamma\eta}+e^{4\gamma\eta}+e^{5\gamma\eta}+e^{6\gamma\eta})\}$$

```
rhs = Simplify@FunctionExpand@Table[QFactorial[n, e^(γ η)], {n, 7}]
```

$\{1,\ 1+e^{\gamma\eta},\ (1+e^{\gamma\eta})(1+e^{\gamma\eta}+e^{2\gamma\eta}),\ (1+e^{\gamma\eta})^2(1+e^{2\gamma\eta})(1+e^{\gamma\eta}+e^{2\gamma\eta}),$
$(1+e^{\gamma\eta})^2(1+e^{2\gamma\eta})(1+e^{\gamma\eta}+e^{2\gamma\eta})(1+e^{\gamma\eta}+e^{2\gamma\eta}+e^{3\gamma\eta}+e^{4\gamma\eta}),$
$(1+e^{\gamma\eta})^3(1+e^{2\gamma\eta})(1-e^{\gamma\eta}+e^{2\gamma\eta})(1+e^{\gamma\eta}+e^{2\gamma\eta})^2(1+e^{\gamma\eta}+e^{2\gamma\eta}+e^{3\gamma\eta}+e^{4\gamma\eta}),$
$(1+e^{\gamma\eta})^3(1+e^{2\gamma\eta})(1-e^{\gamma\eta}+e^{2\gamma\eta})(1+e^{\gamma\eta}+e^{2\gamma\eta})^2$
$(1+e^{\gamma\eta}+e^{2\gamma\eta}+e^{3\gamma\eta}+e^{4\gamma\eta})(1+e^{\gamma\eta}+e^{2\gamma\eta}+e^{3\gamma\eta}+e^{4\gamma\eta}+e^{5\gamma\eta}+e^{6\gamma\eta})\}$

```
lhs == rhs
```

True

```
P[U[g, g, g, g, g], U[h, h, h, h, h]]
```

120

```
P[U[g, g, g, g, g, e, e, e, e], U[h, h, h, h, h, f, f, f, f]] // Factor
```

$120\,(1+e^{\gamma\eta})^2(1+e^{2\gamma\eta})(1+e^{\gamma\eta}+e^{2\gamma\eta})$

## The Double

```
dm_{i_,j_→k_}[ℰ_] := Module[{t1, t2, t3, h1, h2, h3},
  ℰ // qΔ_{i→h1,h2,h3} // S_{h1} // qΔ_{-j→t1,t2,t3} // P_{h1,t1} // P_{h3,t3} // m_{h2,j→k} // m_{-i,t2→-k}]
```

```
U_{-1}[] U_1[g] U_{-2}[h] U_2[] // dm_{1,2→1}
```

$U_{-1}[h]\,U_1[g]$

```
U_{-1}[] U_1[g] U_{-2}[f] U_2[] // dm_{1,2→1}
```

$-\gamma\,U_{-1}[f]\,U_1[\,] + U_{-1}[f]\,U_1[g]$

```
U_{-1}[] U_1[G[1]] U_{-2}[f] U_2[] // dm_{1,2→1}
```

$e^{-\gamma\eta}\,U_{-1}[f]\,U_1[G[1]]$

```
U_{-1}[] U_1[e] U_{-2}[h] U_2[] // dm_{1,2→1}
```

$\eta\,U_{-1}[\,]\,U_1[e] + U_{-1}[h]\,U_1[e]$

```
U_{-1}[] U_1[e] U_{-2}[H[1]] U_2[] // dm_{1,2→1}
```

$e^{\gamma\eta}\,U_{-1}[H[1]]\,U_1[e]$

```
U_{-1}[] U_1[e] U_{-2}[f] U_2[] // dm_{1,2→1}
```

$U_{-1}[H[1]]\,U_1[\,] + U_{-1}[f]\,U_1[e] - U_{-1}[\,]\,U_1[G[1]]$

## The R-Matrix

Using Quesne’s formula.

$$R_{i\_,j\_}[d\_] := \text{Module}\Big[\{x, y\}, \mathbb{O}\Big[U_{-i}[x_1 \to h, x_2 \to f]\, U_i[]\, U_{-j}[]\, U_j[y_1 \to g, y_2 \to e], \text{Series}\Big[\text{Exp}\Big[\hbar\, x_1\, y_1 + \sum_{k=1}^{d} \frac{\left(1 - e^{\hbar\,\eta\,\gamma}\right)^k (\hbar\, x_2\, y_2)^k}{k\left(1 - e^{k\,\hbar\,\eta\,\gamma}\right)}\Big], \{\hbar, 0, d\}\Big]\Big]\Big]$$

$R_{1,2}[4]$

$$U_{-1}[]\, U_2[]\, UU_{-2}[]\, UU_1[] + \left(U_{-1}[f]\, U_2[e]\, UU_{-2}[]\, UU_1[] + U_{-1}[h]\, U_2[g]\, UU_{-2}[]\, UU_1[]\right)\hbar +$$
$$\left(\frac{1}{2} U_{-1}[f, f]\, U_2[e, e]\, UU_{-2}[]\, UU_1[] + U_{-1}[h, f]\, U_2[g, e]\, UU_{-2}[]\, UU_1[] + \frac{1}{2} U_{-1}[h, h]\, U_2[g, g]\, UU_{-2}[]\, UU_1[]\right)\hbar^2 +$$
$$\left(-\frac{1}{4}\gamma\,\eta\, U_{-1}[f, f]\, U_2[e, e]\, UU_{-2}[]\, UU_1[] + \frac{1}{6} U_{-1}[f, f, f]\, U_2[e, e, e]\, UU_{-2}[]\, UU_1[] + \frac{1}{2} U_{-1}[h, f, f]\, U_2[g, e, e]\, UU_{-2}[]\, UU_1[] + \frac{1}{2} U_{-1}[h, h, f]\, U_2[g, g, e]\, UU_{-2}[]\, UU_1[] + \frac{1}{6} U_{-1}[h, h, h]\, U_2[g, g, g]\, UU_{-2}[]\, UU_1[]\right)\hbar^3 +$$
$$\left(-\frac{1}{4}\gamma\,\eta\, U_{-1}[f, f, f]\, U_2[e, e, e]\, UU_{-2}[]\, UU_1[] - \frac{1}{4}\gamma\,\eta\, U_{-1}[h, f, f]\, U_2[g, e, e]\, UU_{-2}[]\, UU_1[] + \frac{1}{24} U_{-1}[f, f, f, f]\, U_2[e, e, e, e]\, UU_{-2}[]\, UU_1[] + \frac{1}{6} U_{-1}[h, f, f, f]\, U_2[g, e, e, e]\, UU_{-2}[]\, UU_1[] + \frac{1}{4} U_{-1}[h, h, f, f]\, U_2[g, g, e, e]\, UU_{-2}[]\, UU_1[] + \frac{1}{6} U_{-1}[h, h, h, f]\, U_2[g, g, g, e]\, UU_{-2}[]\, UU_1[] + \frac{1}{24} U_{-1}[h, h, h, h]\, U_2[g, g, g, g]\, UU_{-2}[]\, UU_1[]\right)\hbar^4 + O[\hbar]^5$$